



TSwap Audit Report

Version 1.0

Lokapal

February 1, 2025

TSwap Audit Report

Lokapal

February 1, 2025

Prepared by:

- Lokapal

Lead Auditor:

- Ricardo Pintos

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Incorrect fee calculation in `TSwapPool::getInputAmountBasedOnOutput` causes protocol to take too many tokens from users, resulting in lost fees

- * [H-2] Lack of slippage protection in `TSwapPool::swapExactOutput`, which allows users to potentially receive fewer tokens
- * [H-3] `TSwapPool::sellPoolTokens` mismatches input and output tokens causing users to receive the incorrect amount of tokens
- * [H-4] In `TSwapPool::_swap`, the extra tokens given to users after every `swapCount` breaks the protocol invariant of $x * y = k$
- Medium
 - * [M-1] `TSwapPool::deposit` is missing deadline check causing transactions to complete even after the user's deadline
- Low
 - * [L-1] `TSwapPool::LiquidityAdded` event parameters out of order
 - * [L-2] `TSwapPool::swapExactInput` does not have a `natspec`
 - * [L-3] Default value returned by `TSwapPool::swapExactInput` results in incorrect return value given
- Informational
 - * [I-1] `PoolFactory::PoolFactory__PoolDoesNotExist` is not used and should be removed
 - * [I-2] Missing zero address check in the `PoolFactory::constructor`
 - * [I-3] `PoolFactory::createPool` should use `.symbol()` instead of `.name()`
 - * [I-4] `PoolFactory::PoolCreated` event does not have indexed parameters
 - * [I-5] Missing zero address check in the `TSwapPool::constructor`
 - * [I-6] `TSwapPool__WethDepositAmountTooLow` is emitting the constant `MINIMUM_WETH_LIQUIDITY`
 - * [I-7] The `TSwapPool::poolTokenReserves` variable is not used:
 - * [I-8] Use of magic numbers
 - * [I-9] `TSwapPool::swapExactInput` function should be marked as `external`
 - * [I-10] Some `public` functions are in the same section as the getter functions
 - * [I-11] Missing `deadline` parameter in `swapExactOutput` `natspec`

Protocol Summary

This project is meant to be a permissionless way for users to swap assets between each other at a fair price. You can think of T-Swap as a decentralized asset/token exchange (DEX). T-Swap is known as an Automated Market Maker (AMM) because it doesn't use a normal "order book" style exchange, instead it uses "Pools" of an asset.

Disclaimer

The LOKAPAL team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: e643a8d4c2c802490976b538dd009b351b1c8dda

Scope

```
1 ./src/  
2 # PoolFactory.sol  
3 # TSwapPool.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- Tokens:
 - Any ERC20 token

Roles

- Liquidity Providers: Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- Users: Users who want to swap tokens.

Executive Summary

This security review was conducted as part of Cyfrin Updraft's Smart Contract Security course. We added 2 findings to the final list of the original project.

Issues found

Severity	Number of issues found
High	4
Medium	1
Low	3
Info	11
Total	19

Findings

High

[H-1] Incorrect fee calculation in TSwapPool::getInputAmountBasedOnOutput causes protocol to take too many tokens from users, resulting in lost fees

Description: The `getInputAmountBasedOnOutput` function is intended to calculate the amount of tokens a user should deposit given an amount of output tokens. However, the function currently miscalculates the resulting amount. When calculating the fee, it scales the amount by 10_000 instead of 1_000.

Impact: Protocol takes more fees than expected from users.

Proof of Concept: Run this test on your `TSwapPool.t.sol` unit suite:

```

1  function testHigherFees() public view {
2      uint256 outputAmount = 10e18;
3      uint256 inputReserves = 100e18;
4      uint256 outputReserves = 100e18;
5      uint256 correctFeesPrecision = 1000;
6      uint256 expectedFees = ((inputReserves * outputAmount) *
7                              correctFeesPrecision) /
8                              ((outputReserves - outputAmount) * 997);
9      uint256 actualFees = pool.getInputAmountBasedOnOutput(
10         outputAmount,
11         inputReserves,
12         outputReserves);
13     assert(actualFees > expectedFees);
14 }

```

Recommended Mitigation: Consider fixing the precision value:

```

1  function getInputAmountBasedOnOutput(
2      uint256 outputAmount,
3      uint256 inputReserves,
4      uint256 outputReserves
5  )
6      public
7      pure
8      revertIfZero(outputAmount)
9      revertIfZero(outputReserves)
10     returns (uint256 inputAmount)
11 {
12 +     return
13 +         ((inputReserves * outputAmount) * 1000) /
14 +         ((outputReserves - outputAmount) * 997);
15 -     return
16 -         ((inputReserves * outputAmount) * 10000) /
17 -         ((outputReserves - outputAmount) * 997);
18 }

```

[H-2] Lack of slippage protection in TSwapPool::swapExactOutput, which allows users to potentially receive fewer tokens

Description: The swapExactOutput function does not include any slippage protection. The TSwapPool::swapExactInput function specifies a minOutputAmount. So, the swapExactOutput should specify a maxInputAmount.

```

1  // IN THE `swapExactInput` FUNCTION

```

```
2 if (outputAmount < minOutputAmount) {
3     revert TSwapPool__OutputTooLow(outputAmount, minOutputAmount);
4 }
```

Impact: If market conditions change before the transaction processes, the user could get a much worse swap.

Proof of Concept: 1. The price of 1 WETH right now is 1,000 USDC, 2. User inputs a swapExactOutput looking for 1 WETH, 1. inputToken = USDC 2. outputToken = WETH 3. outputAmount = 1 4. deadline = 1 minute 3. The function does not offer a maximum input amount, 4. As the transaction is pending in the mempool, the market conditions change, 5. The price moves from 1,000 USDC per WETH to 10,000. 10x more than the user expected, 6. The transaction completes, but the user sent the protocol 10,000 USDC instead of the expected 1,000.

Recommended Mitigation: Consider adding the proper slippage protection and the corresponding custom error:

```
1 // ERRORS
2 + error TSwapPool__InputTooHigh(uint256 actual, uint256 max);
3
4 function swapExactOutput(
5     IERC20 inputToken,
6     IERC20 outputToken,
7     uint256 outputAmount,
8 +     uint256 maxInputAmount,
9     uint64 deadline
10 )
11 public
12 revertIfZero(outputAmount)
13 revertIfDeadlinePassed(deadline)
14 returns (uint256 inputAmount)
15 {
16     uint256 inputReserves = inputToken.balanceOf(address(this));
17     uint256 outputReserves = outputToken.balanceOf(address(this));
18
19     inputAmount = getInputAmountBasedOnOutput(outputAmount, inputReserves,
20         outputReserves);
21 +     if (inputAmount > maxInputAmount) {
22 +         revert TSwapPool__InputTooHigh(inputAmount, maxInputAmount);
23 +     }
24
25     _swap(inputToken, inputAmount, outputToken, outputAmount);
26 }
```

[H-3] TSwapPool::sellPoolTokens mismatches input and output tokens causing users to receive the incorrect amount of tokens

Description: The `sellPoolTokens` function is intended to allow users to easily sell pool tokens and receive WETH in exchange. Users indicate how many pool tokens they're willing to sell in the `poolTokenAmount` parameter. However, the function currently miscalculates the swapped amount.

This is due to the fact that the `swapExactOutput` function is called. But because users specify the exact amount of input tokens, the `swapExactInput` function is the one that should be called.

```
1 function sellPoolTokens(uint256 poolTokenAmount) external returns (uint256
    wethAmount)
2 { return
3     swapExactOutput
4     (
5         i_poolToken,
6         i_wethToken,
7         poolTokenAmount,
8         uint64(block.timestamp)
9     );
10 }
```

Impact: Users will swap the wrong amount of tokens, which is a severe disruption of protocol functionality.

Proof of Concept: Run this test on your `TSwapPool.t.sol` unit suite:

```
1 function testSellPoolTokensReturnsInput() public {
2     vm.startPrank(liquidityProvider);
3     weth.approve(address(pool), type(uint256).max);
4     poolToken.approve(address(pool), type(uint256).max);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6
7     uint256 amountToSell = 1e18;
8
9     uint256 inputReserves = poolToken.balanceOf(address(pool));
10    uint256 outputReserves = weth.balanceOf(address(pool));
11    uint256 amountToSellWithFees = pool.getInputAmountBasedOnOutput(
        amountToSell, inputReserves, outputReserves);
12
13    uint256 ammountToReceiveWithFees = pool.sellPoolTokens(amountToSell);
14    vm.stopPrank();
15    assert(ammountToReceiveWithFees == amountToSellWithFees);
16 }
```

Recommended Mitigation: Consider using the `swapExactInput` function instead of `swapExactOutput`

in the `sellPoolTokens` function. Also, the format of the previous parameters should be modified to match the `swapExactInput` function:

```
1 function sellPoolTokens(  
2     uint256 poolTokenAmount,  
3 +   uint256 minOutputToken  
4 ) external returns (uint256 wethAmount)  
5 { return  
6 +     swapExactInput  
7 -     swapExactOutput  
8     (  
9         // CORRECT `swapExactInput` FORMAT  
10 +     i_poolToken,  
11 +     poolTokenAmount,  
12 +     i_wethToken,  
13 +     minOutputToken,  
14 +     uint64(block.timestamp)  
15 -     i_poolToken,  
16 -     i_wethToken,  
17 -     poolTokenAmount,  
18 -     uint64(block.timestamp)  
19     );  
20 }
```

[H-4] In `TSwapPool : : _swap`, the extra tokens given to users after every `swapCount` breaks the protocol invariant of $x * y = k$

Description: The protocol follows a strict invariant of $x * y = k$, where: - x is the balance of the pool token, - y is the balance of WETH, - k is the constant product of the two balances.

This means that, whenever the balances change in the protocol, the ratio between the two amounts should remain constant, hence the k . However, this is broken due to the extra incentive in the `_swap` function. Meaning that overtime the protocol funds will be drained.

```
1 swap_count++;  
2 if (swap_count >= SWAP_COUNT_MAX) {  
3     swap_count = 0;  
4     outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);  
5 }
```

Impact: A user could maliciously drain the protocol of funds by doing multiples swaps and collecting the extra incentive given out by the protocol. This breaks the invariant k .

Proof of Concept: Run this test on your `TSwapPool.t.sol` unit suite:

```
1 function testInvariantBroken() public {
2     vm.startPrank(liquidityProvider);
3     weth.approve(address(pool), 100e18);
4     poolToken.approve(address(pool), 100e18);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6     vm.stopPrank();
7
8     uint256 outputWeth = 1e17;
9     vm.startPrank(user);
10    poolToken.mint(user, 100e18);
11    poolToken.approve(address(pool), type(uint256).max);
12    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
13    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
14    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
15    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
16    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
17    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
18    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
19    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
20    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
21
22    int256 startingY = int256(weth.balanceOf(address(pool)));
23    int256 expectedDeltaY = int256(-1) * int256(outputWeth);
24    // The 10th swap gives the incentive, breaking the invariant
25    pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
26    vm.stopPrank();
27
28    uint256 endingY = weth.balanceOf(address(pool));
29    int256 actualDeltaY = int256(endingY) - int256(startingY);
30    // This assertion will fail
31    assertEq(actualDeltaY, expectedDeltaY);
32 }
```

Recommended Mitigation: Consider removing the extra incentive. Otherwise, the protocol should account for the change in the invariant after each 10th swap.

```
1 - swap_count++;
2 - if (swap_count >= SWAP_COUNT_MAX) {
3 -     swap_count = 0;
4 -     outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);
5 - }
```

Medium

[M-1] TSwapPool::deposit is missing deadline check causing transactions to complete even after the user's deadline

Description: The deposit function accepts a deadline parameter, which according to the documentation is “The deadline for the transaction to be completed by”. However, this parameter is never used. As a consequence, operations that add liquidity to the pool might be executed at unexpected times, at market conditions where the deposit rate is unfavorable.

Impact: Transactions could be sent when market conditions are unfavorable to deposit, even when adding a deadline parameter.

Proof of Concept: This is the terminal message when running `forge build` on this project:

```
1 Compiler run successful with warnings:
2 Warning (5667): Unused function parameter. Remove or comment out the variable
   name to silence this warning.
3 --> src/TSwapPool.sol:117:9:
4     |
5 117 |         uint64 deadline
6     |         ^^^^^^^^^^^^^^^^^
```

Recommended Mitigation: Consider adding the `revertIfDeadlinePassed` modifier:

```
1     function deposit(
2         uint256 wethToDeposit,
3         uint256 minimumLiquidityTokensToMint,
4         uint256 maximumPoolTokensToDeposit,
5         uint64 deadline
6     )
7     external
8     revertIfZero(wethToDeposit)
9 +    revertIfDeadlinePassed(deadline)
10    returns (uint256 liquidityTokensToMint)
```

Low

[L-1] TSwapPool::LiquidityAdded event parameters out of order

Description: There is a mismatch between the parameters in the event `LiquidityAdded`:

```
1     event LiquidityAdded(
2         address indexed liquidityProvider,
```

```
3         uint256 wethDeposited,  
4         uint256 poolTokensDeposited  
5     );
```

And the emitted `LiquidityAdded` in the `_addLiquidityMintAndTransfer` function:

```
1     emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit);
```

Impact: When querying the event logs, users will have the information mixed up.

Proof of Concept:

1. User adds liquidity,
2. User queries the amount of `weth` deposited,
3. User actually gets the amount of `poolTokens` deposited.

Recommended Mitigation: Consider fixing the `emit` to match the event parameters order:

```
1 +   emit LiquidityAdded(msg.sender, wethToDeposit, poolTokensToDeposit);  
2 -   emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit);
```

[L-2] TSwapPool : : swapExactInput does not have a natspec

Description: The `swapExactInput` function does not have a natspec. This can leave users without proper information about the function that they are calling. Additionally, other important functions have their respective natspec.

Recommended Mitigation: Consider adding the natspec for the `swapExactInput` function. This is a suggestion:

```
1  /*  
2  * @notice swapExactInput will swap the input token for the output token  
3  * @param inputToken the token the user is swapping  
4  * @param inputAmount the amount the user is swapping  
5  * @param outputToken the token the user is receiving  
6  * @param minOutputAmount the minimum amount the user wants to receive  
7  * @param deadline the deadline for the transaction to be completed by  
8  */
```

[L-3] Default value returned by TSwapPool : : swapExactInput results in incorrect return value given

Description: The `swapExactInput` function is expected to return the actual amount of tokens bought by the caller. However, while it declares the named return value `output` it is never assigned a value,

nor uses an explicit return statement.

Impact: The returned value will always be zero, giving incorrect information to the caller.

Proof of Concept: Run this test on your TSwapPool.t.sol unit suite:

```
1 function testSwapExactInputReturnsZero() public {
2     vm.startPrank(liquidityProvider);
3     weth.approve(address(pool), 100e18);
4     poolToken.approve(address(pool), 100e18);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6     vm.stopPrank();
7
8     vm.startPrank(user);
9     poolToken.approve(address(pool), 10e18);
10    uint256 expected = 9e18;
11
12    uint256 shouldNotBeZero = pool.swapExactInput(poolToken, 10e18, weth,
13        expected, uint64(block.timestamp));
14    vm.stopPrank();
15    assert(shouldNotBeZero == 0);
16 }
```

Recommended Mitigation: Consider replacing outputAmount for output on the swapExactInput function:

```
1 {
2     uint256 inputReserves = inputToken.balanceOf(address(this));
3     uint256 outputReserves = outputToken.balanceOf(address(this));
4 +   uint256 output = getOutputAmountBasedOnInput
5 -   uint256 outputAmount = getOutputAmountBasedOnInput
6     (
7         inputAmount,
8         inputReserves,
9         outputReserves
10    );
11
12 +   if (output < minOutputAmount){revert TSwapPool__OutputTooLow(output,
13     minOutputAmount);}
13 -   if (outputAmount < minOutputAmount){revert TSwapPool__OutputTooLow(
14     outputAmount, minOutputAmount);}
15
16 +   _swap(inputToken, inputAmount, outputToken, output);
17 -   _swap(inputToken, inputAmount, outputToken, outputAmount);
18 }
```

Informational

[I-1] PoolFactory::PoolFactory__PoolDoesNotExist is not used and should be removed

Description: The PoolFactory::PoolFactory__PoolDoesNotExist error is not used in the contract:

```
1      error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

Recommended Mitigation: Consider removing the error or implementing it in the corresponding section:

```
1 -    error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[I-2] Missing zero address check in the PoolFactory::constructor

Description: The constructor does not have a zero address check on the address argument:

```
1      constructor(address wethToken) {
2          i_wethToken = wethToken;
3      }
```

Recommended Mitigation: Consider adding a zero address check and the corresponding custom error:

```
1      // ERRORS
2 +    error PoolFactory__NotZeroAddress();
3
4      // FUNCTIONS
5      constructor(address wethToken) {
6 +    if (address(wethToken) != address(0)) {
7 +        i_wethToken = wethToken;
8 +    } else {
9 +        revert PoolFactory__NotZeroAddress();
10 +    }
11 -        i_wethToken = wethToken;
12     }
```

[I-3] PoolFactory::createPool should use .symbol() instead of .name()

Description: The PoolFactory::createPool function uses the .name() to complete the LT symbol, but there is already the .symbol() function, which returns a more appropriate reference and is a

shorter string than the token name:

```
1      string memory liquidityTokenSymbol = string.concat("ts", IERC20(
      tokenAddress).name());
```

Recommended Mitigation: Consider replacing `.name()` for `.symbol()`:

```
1 +   string memory liquidityTokenSymbol = string.concat("ts", IERC20(
      tokenAddress).symbol());
2 -   string memory liquidityTokenSymbol = string.concat("ts", IERC20(
      tokenAddress).name());
```

[I-4] PoolFactory::PoolCreated event does not have indexed parameters

Description: The `PoolFactory::PoolCreated` event has two address parameters, but none are flagged as **indexed**:

```
1      event PoolCreated(address tokenAddress, address poolAddress);
```

Recommended Mitigation: Consider adding the **indexed** flag to both parameters, which will make them easier to query by off-chain resources:

```
1 +   event PoolCreated(address indexed tokenAddress, address indexed poolAddress
      );
2 -   event PoolCreated(address tokenAddress, address poolAddress);
```

[I-5] Missing zero address check in the TSwapPool::constructor

Description: The **constructor** does not have a zero address check on the address argument:

```
1      constructor(
2          address poolToken,
3          address wethToken,
4          string memory liquidityTokenName,
5          string memory liquidityTokenSymbol
6      ) ERC20(liquidityTokenName, liquidityTokenSymbol) {
7          i_wethToken = IERC20(wethToken);
8          i_poolToken = IERC20(poolToken);
9      }
```

Recommended Mitigation: Consider adding a zero address check and the corresponding custom error:

```
1      // ERRORS
```

```
2 + error TSwapPool__NotZeroAddress();
3
4 // FUNCTIONS
5 constructor(
6     address poolToken,
7     address wethToken,
8     string memory liquidityTokenName,
9     string memory liquidityTokenSymbol
10 ) ERC20(liquidityTokenName, liquidityTokenSymbol) {
11 +     if (address(wethToken) != address(0) && address(poolToken) != address
12     (0)) {
13 +         i_wethToken = IERC20(wethToken);
14 +         i_poolToken = IERC20(poolToken);
15 +     } else {
16 +         revert PoolFactory__NotZeroAddress();
17 -     }
18 -     i_wethToken = IERC20(wethToken);
19 -     i_poolToken = IERC20(poolToken);
20 }
```

[I-6] TSwapPool__WethDepositAmountTooLow is emitting the constant MINIMUM_WETH_LIQUIDITY

Description: The TSwapPool__WethDepositAmountTooLow error has the parameter `minimumWethDeposit` :

```
1 error TSwapPool__WethDepositAmountTooLow(
2     uint256 minimumWethDeposit,
3     uint256 wethToDeposit
4 );
```

This error is used only in the deposit function, taking the constant `MINIMUM_WETH_LIQUIDITY` as an argument.

```
1 if (wethToDeposit < MINIMUM_WETH_LIQUIDITY) {
2     revert TSwapPool__WethDepositAmountTooLow(
3         MINIMUM_WETH_LIQUIDITY,
4         wethToDeposit
5     );
6 }
```

That constant can already be accessed with the `getMinimumWethDepositAmount` getter function, in case the user needs that information. This makes the revert less gas efficient.

Recommended Mitigation: Consider removing the `minimumWethDeposit` as a parameter from

the TSwapPool__WethDepositAmountTooLow error, therefore removing the need to add the MINIMUM_WETH_LIQUIDITY constant:

```
1 // ERRORS
2 error TSwapPool__WethDepositAmountTooLow(
3 -     uint256 minimumWethDeposit,
4     uint256 wethToDeposit
5 );
6
7 // DEPOSIT FUNCTION
8 if (wethToDeposit < MINIMUM_WETH_LIQUIDITY) {
9     revert TSwapPool__WethDepositAmountTooLow(
10 -         MINIMUM_WETH_LIQUIDITY,
11         wethToDeposit
12     );
13 }
```

[I-7] The TSwapPool::poolTokenReserves variable is not used:

Description: This line in the deposit function has a variable that is not used:

```
1 uint256 poolTokenReserves = i_poolToken.balanceOf(address(this));
```

This issue even appears when building the project:

```
1 Warning (2072): Unused local variable.
2 --> src/TSwapPool.sol:131:13:
3   |
4 131 |         uint256 poolTokenReserves = i_poolToken.balanceOf(address(
5   |         this));
```

Recommended Mitigation: Consider removing it from the function:

```
1 - uint256 poolTokenReserves = i_poolToken.balanceOf(address(this));
```

[I-8] Use of magic numbers

Description:

getOutputAmountBasedOnInput

```
1 uint256 inputAmountMinusFee = inputAmount * 997;
2     uint256 numerator = inputAmountMinusFee * outputReserves;
3     uint256 denominator = (inputReserves * 1000) + inputAmountMinusFee;
```

```
getInputAmountBasedOnOutput return ((inputReserves * outputAmount) * 10000) / ((outputReserves - outputAmount) * 997);
```

Recommended Mitigation:**[I-9] TSwapPool : swapExactInput function should be marked as external**

Description: The swapExactInput function is marked `public` but is not used internally. This increase gas cost in the deployment.

Recommended Mitigation: Consider using the `external` flag for the swapExactInput function.

[I-10] Some public functions are in the same section as the getter functions

Description: The getPoolTokensToDepositBasedOnWeth and totalLiquidityTokenSupply functions are grouped with the getter functions.

Recommended Mitigation: Consider adding a `// PUBLIC VIEW //` section for those functions.

[I-11] Missing deadline parameter in swapExactOutput natspec

Description: The swapExactOutput function does not have the corresponding natspec information for the deadline parameter.

Recommended Mitigation: Consider adding the same natspec line of the swapExactInput for the deadline parameter:

```
1      /* @param deadline the deadline for the transaction to be completed by
```